
SUMMARY 

Concrete weight coating on subsea pipelines is typically not continuous over the field 
joints.  This can result in significant strain concentrations in the field joint area, especially 
when the pipe is subject to plastic deformations and strain-based design is used.  The 
reason for this is the extra bending stiffness provided by the concrete.  As a result imposed 
bending deformations (e.g. due to the pipe going over the stinger of the laybarge, or to 
accommodate thermal strains by lateral buckling) tend to concentrate in the rather short 
field joint. 

To address this without exaggerating the stiffening effect of the concrete, it is necessary to 
account for slip between the concrete weight coating and the pipe.  For this purpose a 
model developed by STATOIL is implemented.  The advantage of this implementation 
over that in STATOIL’s MomKap program is that it allows the effect of concrete coating 
to be included in any finite element analysis using beam-type elements for the pipe, 
including geometric as well as material nonlinearities.  Thus, for instance, concrete coating 
can be included in a model for lateral buckling of a pipeline.  However the current 
implementation is 2-dimensional.  The pipe must remain on one plane. 

The approach is based on using 4 degrees of freedom per node: in addition to 2 
displacement components and one rotation the relative slip displacement between the and 
the pipe and the concrete is introduced as the 4th degree of freedom. 

This report includes a description of the formulation including assumptions made and 
validation of the implementation against results that have been provided by STATOIL. 

1. THEORY AND IMPLEMENTATION 

A special finite element as been developed to account for the effect of discontinuous 
concrete coating on the behaviour of pipes in the non-linear range according to the 
methodology of Verley and Ness [1][2].  The general modular finite element program 
(npex) developed at the University of Michigan, was used as the framework within which 
to develop this element.  This program already contains two-dimensional pipe elements, 
which can be used together with the new concrete elements to model the interaction 
between the pipe and the coating. 

Essentially the formulation accounts for the stiffness of the concrete in compression as 
well as the steel in tension or compression.  In tension the concrete is assumed to be 
cracked , so that the stress is zero.  However due to axial slip of the concrete over the pipe, 
the strain in the concrete is not that same as that in the pipe.   

Consider the pipe in the X-Z plane, where X and Z are fixed cartesian coordinates, with  X 
being the axial coordinate and Z the transverse coordinate.  The concrete and the steel can 
be envisioned as separate beam type structures.  Both are treated using the moderate 
deflection beam theory, including shear deformations.  The transverse Z component of 
displacement w=w(X) and rotation θ=θ(X) is assumed to be the same for both the 
concrete and the pipe.  However the axial displacements u=u(X) are different for the steel 
and the concrete.  The X-component of the force which the concrete exerts on the beam, 
depends on the difference in the axial displacements, 

∆u(X) = upipe(X) – uconcrete(X) (1.1) 
according to an elastic-perfectly plastic type relationship, based on the maximum shear 
strength τy of the tar, and a mobilisation displacement needed to develop this maximum 
shear strength typically taken ∆uy=2mmi.  In addition to the axial force transfer according 
to this elastic perfectly plastic relationship, transverse forces (in the Z direction) and body 
couples are also transferred between the steel and the concrete, as needed to ensure 
equality of w(X) and θ(X). 

For the finite element formulation in the X-Z plane, four degrees of freedom per node are 
used: axial displacement u  for the pipe, transverse displacement w, rotation θ, and axial 
slip ∆u the concrete relative to the steel. 

The concrete is assumed to resist axial strains in compression only according to a parabolic 
stress-strain relationship given by 

fc   =    fc’  { 1 – [(εc’ – εc)/εc’]
2 } (1.2) 

in which 

fc = axial stress in the concrete (positive for compression) 

fc’ = compressive strength of concrete (typically around 40MPa) 

εc = axial strain in the concrete (positive for compression) 

εc’ = strain at which the compressive strength of the concrete is reached (typically 0.2% 
used) 

                                                 
i  Since the slip is typically well into the yielded region, this parameter has little influence on the results.  However a 

finite value must be chosen to prevent ill-conditioning of the calculations and/or dividing by zero. 

The above relation applies until the compressive strength of the concrete is reached.  
Beyond that point the concrete tends to loose strength.  However to avoid convergence 
difficulties it is assumed that the stress in the concrete remains constant, if it is deformed 
beyond a strain of εc’.

ii  Under extension, the stress in the concrete is taken to be zero, due 
to cracking of the concrete. 

In the npex program the concrete coating is implemented as a separate element according 
to a unified modular structure for adding new elements in npex.  The routines are included 
in Appendix B, and the interface between these routines and the npex program is described 
in Appendix A.  The model constructed has 4 degrees of freedom per node.  The nodes are 
connected by pipe elements, and also by concrete coating elements where the pipe is 
coated.  A node is always required where the coating ends (typically around 300mm to each 
side of the girth weld).  For the nodes within a not-coated part of the pipeline (e.g. in the 
middle of the field joint area), and other nodes that do not have any concrete coating 
element attached to them, the 4th degree of freedom needs to be eliminated (by setting the 
corresponding displacement to zero). 

                                                 
ii  If the calculated maximum compressive strain in the concrete is larger than εc’, this means that crushing of the 

concrete occurs.  This is regarded as a situation to be avoided by proper design, rather than analysed.  To analyse it, 
one would need to modify this model to include the drop in concrete resistance upon crushing. 

2. VALIDATION OF NPEX IMPLEMENTATION 

STATOIL (Erik Levold) kindly provided a verification example calculated with their 
MOMKAP software, which embodies the implementation of the methods of Ness and 
Verley [1] described above.  The problem solved represents a long string of 12.2m-long 
pipe joints under uniform bending.  The concrete continues up to 35cm from the field 
joint.  Thus at each field joint there is a 70cm-long section of pipe that is not coated with 
concrete.  Additional parameters of this validation problem are as follows: 

Pipe outer diameter D = 1078 mm 

Pipe wall thickness tpipe =  30.8 mm 

Tar thickness ttar  =  6 mm 

Concrete coating thickness tconc = 45 mm 

Concrete compressive strength fc’ = 40 MPa 

Concrete strain at ultimate εc’ = 0.2% 

Tar Shear Strength τy = 0.1 MPa 

In view of the symmetries involved in a long string of pipe under pure bending (i.e. no 
axial load or shear force throughout), it is sufficient to model a section of pipe from the 
middle of a field joint, to the middle of a pipe joint, as shown in Figure 2.1.  The pipe only 
to applied bending moment with no axial or shear forces, so it does not matter which end 
is fixed.  In this case fixing the pipe at the field joint was more convenient for post-
processing of the results. 

For the stress-strain relation of the steel some curve fitting was needed, because the 
MOMKAP software uses a different form of the stress-strain curve, that that used by the 
pipe element in npex.  Specifically the MOMKAP  results are based on a Tvergaard type 
relationship given by 

ε  =  (σy/E) {[(σ/σy)
n – 1]/n + 1} (2.1) 

with E = 207 GPa, σy = 372.6 MPa and n = 16.26.  On the other hand the NPEX results 
are based on a Dafalias stress-strain curve [3].  By choosing the Dafalias material 
parameters as 

Initial yield stress  σ0 = 372.6 MPa 

Bounding stress σb = 492 MPa 

Hardening modulus h = 20,000 MPa 

Slope of the Bounding line E0
p = 1250 MPa 

good agreement between the stress strain curves of the two models was achieved, as shown 
in Figure 2.2. 

The approximately matching Tvergaard and Dafalias stress-strain curves are chosen here 
for comparison and validation purposes only.  In actual applications it is important to 
choose the stress-strain curve the best describes the actual behaviour of the material, 
because the degree of strain concentration can be quite sensitive to the shape of the stress 
strain curve.  Generally at strains above yield a flat stress strain curve (e.g. an elastic-
perfectly plastic curve to represent Lüder’s banding) leads to higher strain concentrations.  
However the rounded stress strain curves used here involve incipient yielding at stresses 

below the API yield stress (defined as the stress at 0.5% strain).  Therefore at low strains, 
such as those on the stinger for pipelay in shallower water, an elastic perfectly plastic stress-
strain curve gives rise to lower strain concentrations than those reported here for the 
Tvergaard or Dafalias stress-strain curves. 

The comparison of the results provided by STATOIL from MOMKAP with those from 
the newly developed element in NPEX are shown in Figure 2.3 to Figure 2.6.  A more 
precise definition of the terminology used therein is as follows: 

• The “nominal bending strain” is defined as the maximum strain calculated from the 
curvature using elementary beam theory.  In all cases it is calculated for the outer 
surface of the pipe.  (Thus the nominal bending strain is equal to the curvature 
multiplied by one half the outer diameter of the steel pipe.) 

• The “global curvature” is the relative rotation over the whole model divided by the 
length of the whole model.  This is also equal to the average curvature over the length 
of the model. 

• The “global bending strain” or “global nominal bending strain” or “average bending 
strain” is half the outer diameter of the steel pipe multiplied by the global curvature. 

It is clear from the comparisons in Figure 2.3 to Figure 2.6 that the agreement is as good as 
might be expected given slight differences in the stress strain curves, in the discretisation or 
slightly different quantities plotted (e.g. nominal bending strain vs. actual strain).  

Figure 2.7 and Figure 2.8 are provided to further illustrate the behaviour of the concrete 
coated pipe, and verify that the solutions are reasonable. 

In Figure 2.7 it is seen that the axial force in the concrete increases more or less linearly 
with the distance from the field joint.  This is due the shear stress τy transmitted across the 
tar coating between the concrete and the steel, due to the tendency for the coating to slip 
axially towards the field joint.  The axial force in the concrete is compressive, and it is 
balanced by an equal and opposite force in the steel pipe, so that the net axial force is zero, 
as required from equilibrium. 

Since the concrete cannot carry any tension, it must rely on a compressive axial force in 
order to be able to carry any bending moment.  It is seen from Figure 2.8 that this results 
in a bending moment in the concrete that approximately proportional to the axial force of 
Figure 2.7.   The total bending moment (from steel + concrete) is constant over the length 
of the pipe, as it must be for pure bending, from equilibrium considerations.  

It is also clear from Figure 2.8 that the bending moment carried by the concrete is a small 
fraction of the total.  Nevertheless, on the flat portions of the moment-curvature 
relationships, this can give rise to a considerable variation on the curvatures along the 
length of the pipe. 

Figure 2.1: Finite element model used (Lrep is taken as 6.1m, i.e. half the joint length) 

 

 

 

0

100

200

300

400

500

600

0.0% 0.5% 1.0% 1.5% 2.0% 2.5% 3.0%
Strain

S
tre

ss
 (M

P
a)

Tvergaard Relation (MOMKAP)
Dafalias Model (npex)
Dafalias Bounding Line

 

Figure 2.2: Comparison of the stress-strain curves used in the validation analyses. 

Figure 2.3: Comparison of Strain concentration factors (defined as the ratio of the 
curvature in the field joint, divided by the global average curvature). 

Figure 2.4: Comparison of moment curvature relationships. 

Figure 2.5: Comparison of concrete axial slip at the field joint. 

Figure 2.6: Comparison of axial strain distributions along the length of the pipe, and a 
global nominal bending strain of 1%.  (Note that npex results represent 
actual strains in the steel at the midsurface, whereas those from MOMKAP 
are nominal strains based on the curvature.  The tensile strain in the steel is 
a bit higher than the compressive strain at the opposite side, some 
compression is carried by the concrete.) 

Figure 2.7: Distribution of axial force (the compressive force in the concrete at any 
point is equal in magnitude to the tensile force in the steel).  The linear 
variation in axial force arises because of a constant transfer of axial force per 
unit length from the steel to the concrete determined by the yield strength 
of the tar interface. 

Figure 2.8: Distribution of bending moments along the length of the half-joint from 
npex calculation.  (Note that the total bending moment is exactly constant, 
as it must be from equilibrium under conditions of pure bending.) 

3. ESTIMATION OF THE EFFECTIVE CONCRETE-PIPE BOND SHEAR 
STRENGTH 

Perhaps the most difficult model parameter to estimate is the effective shear strength of 
the bond between the concrete and the corrosion coating.  Tests for this reported in [2] 
give the following shear capacities: 

• 0.42 to 0.55MPa for asphalt enamel corrosion coating at room temperature 

• 0.28 to 0.35MPa for Polyethene corrosion coating. 

However complicating factors arise because the bond does not behave as an elastic 
perfectly plastic material, as is assumed in the analysis.  These differ depending on the type 
of corrosion coating. 

It must be borne in mind that the effective bond stress is not only a material property but 
also can compensate for assumptions made that are not an exact description of the real 
behaviour.  One such assumption that the rotations of cross sections of the concrete are 
the same as those for the pipe, as illustrated in Figure 3.3a.  I.e. it is assumed that the 
relative movement of the pipe and the concrete is purely axial, and is associated with equal 
axial slip all around the circumference.  In reality the driving force for such axial slip comes 
from the compression side only.  This tends to make cross sections of the concrete rotate 
with respect to the pipe, as shown in Figure 3.3b, so that more axial slip occurs on the 
compression side than on the tension side.  The consequences of this behaviour are as 
follows: 

a) At low strains there may not be enough axial slip on the tension side to develop the 
resistance τY there.  Thus the effective shear strength τY is lowered, because the 
shear strength is not developed over the entire circumference. 

b) At higher strains, the relative rotation can become so large that rings of the 
concrete tend to jam, thereby developing an extra frictional axial resistance.  This 
can be enhanced by radial expansion of the pipe due to internal pressure.  Of 
course significant jamming forces can only be developed as a result of the hoop 
reinforcement in the concrete. 

3.1. Asphalt Corrosion Coating 

Asphalt corrosion coating behaves as a visco-plastic material as illustrated by the test 
results from [2] reproduced in Figure 3.1.  The comparison in [2] between the full scale 
bending tests to the analysis results lead to good agreement using a shear strength of 
0.55MPa for rapid loading (Test 2), but for slower loading (over a period of  80min, Test 1) 
the good match was obtained for a shear strength of around 0.2 to 0.25MPa.  This is 
consistent with the bond test results in Figure 3.1.  

Thus for design based on the assumption of elastic perfectly-plastic behaviour, a bond 
strength around 0.5MPa seems appropriate for rapid loading (e.g. as the pipe enters the 
stinger), and 0.25MPa seems more appropriate for slower loading such as lateral buckling 

where it can be shown that dynamiciii buckling either does not occur, or occurs at a 
sufficiently lowiv temperature. 

The bond strength should be reduced for the effect of temperature.  E.g. for slow loading 
at temperatures around 50°C a shear strength as low as 0.1MPa may be appropriate.  
Indeed as the temperature approaches the limit for application of asphalt of around 60°C 
to 70°C the bound strength can be expected to become very small, with the asphalt 
essentially providing lubrication rather than resistance. 

3.2. Polyethene or Polypropylene Coating 

The test results for polyethene (reproduced from [2] in Figure 3.2) show brittle rather than 
viscous behaviour.  The maximum resistance is reached at small displacements (ranging 
from 0.1mm to 0.35mm), after which the shear stress drops rapidly to a residual value of 
0.1MPa, “due mainly to friction” [2]. 

The questions that then arise are “At what point in the loading of the pipe will this brittle 
fracture of the bond occur?” and “Could significant strain concentrations in the steel 
develop before brittle fracture of the bond?”  Fortunately in absence of slip the bond shear 
stress is fully concentrated at the point were the concrete stops, i.e. at the field joint.  
Therefore the bond is soon broken there and debonding can then propagate further into 
the concrete-coated portion of the pipe.  Thus the shear stress for most of the interface 
will be at the residual value of 0.1MPa.  

The above suggests that 0.1MPa is an appropriate value for use in design based on elastic-
perfectly plastic behaviour.  However it must be borne in mind that the frictional effects 
may not always be the same as for the tests performed in [2].  E.g. the normal force at the 
interface may increase due to pressurisation of the pipe, or jamming could occur as rings of 
concrete coating tend to twist with respect to the pipe, as illustrated in Figure 3.3.  Indeed 
the full scale test in [2] showed a bit higher strain concentrations than those predicted with 
a shear strength of 0.1MPa (see Test 4, Figure 15 in [2]), but it is difficult to assess exactly 
to what extent this is due to the concrete cover, or due to local variations in steel strength. 

In view of the above 0.2MPa is considered to be a prudent design value for the bond 
strenght based on the assumption of elastic-perfectly plastic behaviour.  This can 
reasonably be applied for polypropylene coating as well as polyethene.  With further 
investigation involving an assessment of the mechanisms that lead to the residual frictional 
resistance a lower value may be justified.  

In lieu of test data on the behaviour of the interface at higher temperatures, the behaviour 
of the interface between the polyethylene or polypropylene coating and the concrete may 
be assumed to be temperature independent.  Indeed although the brittle resistance part of 
the bonding may be affected, this part of the resistance hardly affects the behaviour, and 
the later frictional resistance is less likely the be affected, unless the temperatures rise to the 
vicinity of the melting point for the polyethylene or polypropylene. 

                                                 
iii  Dynamic buckling can occur if the initial imperfections to trigger buckling are small, so that the axial force rises 

considerably before buckling occurs. 

iv  There “low” means low enough so that the pipe deformations at this temperature will not be dangerous even for a 
high effective shear strength of the bond. 

Figure 3.1: Results of bond shear strength tests for asphalt corrosion coating at room 
temperaturev from [2].  Shows shear stress on the interface (MPa) as a 
function of the relative displacement on the interface in mm.  Different 
loading histories were applied, including monotonic tests at different rates 
(A1, A2, A3), creep tests in which the load was held constant (A4 and A5), 
and a relaxation test (A6) in which the displacement was held nominally 
constant (except for flexibility in the test setup) at various levels. 

                                                 
v The test temperature is not reported in [2], and is therefore presumed to be at room temperature. 



EP 2006-5161 - 12 - Unrestricted 
 

 

Figure 3.2: Bond shear strength test results for polyethene corrosion coating from [2].  
The bond stress drops rapidly from a maximum value tm to a residual value 
of tresidual = 0.1MPa, which is thought to be due to friction. 

 

 

Figure 3.3: Top: idealised deformation mechanism for concrete with plane 
perpendicular sections in the concrete remaining plane and perpendicular 
to the axis of the pipeline, but sliding axially along the pipeline because of 
the high stiffness of the concrete in compression relative to the bond 
strength.  Bottom: actual behaviour – the rotation of the concrete sections 
does not quite match that of the pipe, so that axial sliding is more on the 
compression side that on the tension side.  This type of behaviour is evident 
from Figure 12 of [2].  As a result of this, the concrete rings may have a 
tendency to jam, thereby increasing the effective axial resistance.  This is 
because the frictional part of the resistance with rise with increasing normal 
force due to jamming 

APPENDIX A. DESCRIPTION NPEX ELEMENT ROUTINES 

The finite element calculations reported on here were performed by adding two new types 
of elements to a general finite element code.  For each element type there is a set of two 
element routines: one by which material properties etc. are read and the storage 
requirements of the element are defined, and one to perform the actual calculations of the 
element contribution of the effective out-of-balance nodal forces and the element 
contribution to the tangent stiffness matrix. This Appendix describes the requirements for 
these two routines. 

The routine to read material properties etc. is named “RELPROP_nn” where “nn” stands 
for a two digit number describing the element type, and the routine to perform the 
required calculations is named “GENELKP_nn”.  A description of the arguments for 
these routines and how they are to be used or defined is given in Tables A1 and A2.   

Element groups are used, with all elements within a group being of the same type and 
having the same properties.  These element properties are stored in an array AELPROP or 
IELPROP depending on whether they are real- or integer-valued. The RELPROP_nn 
routine is called once for every element group at the beginning of the analysis.  Its function 
is the read the element properties. Typical element properties include material properties 
such at Young’s modulus, the yield stress, and other material parameters as might be 
required by the material model used.  They can be treated as element properties as long as 
the same values apply for all elements in the group.  In contrast other quantities, such as 
the current stress in the element, or the current location and size of the yield surface vary 
from one element to the next within an element group.  These will be referred to as 
“element parameters”.  They must be stored separately for each element and will typically 
also be updated at every loadstep once a converged solution has been obtained, before 
proceeding to the next loadstep. 

The key output from the GENELKP routine are the out-of-balance forces ELP and the 
element tangent stiffness matrix ELK, both of which are assembled into corresponding 
global out-of-balance forces and the global tangent stiffness matrix used in the solution 
process by Newton iteration. 

The program uses a simple user interface by which (unit 5) standard input is entered in 
response to prompts generated by the main program or the element routine 
RELPROP_nn.  The input is then written to unit 9.  One can then repeat the analysis 
without re-entering interactively all the input data, and make changes in some of the input 
parameters, if desired.  Also if one has completed only part of an interactive input session, 
one can interrupt, correct values entered earlier as needed using the unit 9 file. Upon 
restarting the program will then first use that data available in unit 9, and then prompt the 
user interactively for the remainder of the required data.  Details can be found by 
examining and of the ReaD RECord (RDREC) routines listed in Appendix C. 

To describe the loading, load functions ALOADF may be used from within the element 
routine.  Specifically ALOADF(TIME,i,n) provides the nth time derivative of the ith load 
factor as a function of a load parameter named TIME.  In the analyses reported herein four 
such load functions were used: one for the initial curvature history to describe reeling, one 
for the submerged weight of the pipe, one for the internal pressure, and one for 
temperature.  These load functions are defined outside the element routines by linear 
interpolation between user-specified values.  The data describing the load functions are 
entered as part of the input data (outside the element routine), and stored in a common 
block.  Therefore they need not pass through the element routines.  Periodic load functions 
may also be used if one wants to examine ratcheting. 

Table A.1: Arguments to subroutine RELPROP_nn. 

Name Description Input/ 
Outpu
t 

AELPROP Array containing real-valued element properties to be read by  the 
RELPROP routine either from unit 9 or from unit 5 depending on 
the value of IU9STAT. 

O 

IELPROP Array containing integer valued element properties that are to be 
read by the RELPROP routine either from unit 9 or from unit 5 
depending on the value of IU9STAT. 

O 

IEXFLAG Contains information about type of analysis being performed, and 
may be needed only for special elements (e.g. to give an indication 
to read some additional element properties when needed) 

I 

IU9STAT If IU9STAT=0 input data should be read from unit 9; otherwise 
they should be read from unit 5 and written to unit 9.  If an 
attempt to read some input from unit 9 fails, then IU9STAT 
should be reset to a non-zero value and the required input should 
be read from unit 5 instead (preferably after prompting the user for 
the input), and written to unit 9.  This should be done for every 
record that is read and has been achieved here by routines whose 
name starts with “RDREC”, which are listed in Appendix C. 

I & O 

NAELDAT
1 

Maximum length of AELPROP (usually a very large number, but 
may be used to check that available storage for AELPROP is not 
exceeded). 

I 

NIELDAT1 Maximum length of IELPROP (usually a very large number, but 
may be used to check that available storage for IELPROP is not 
exceeded). 

I 

NEN Number of nodes the element will be attached to. O 
NAELPAR Number of real-valued element parameters used for each element.  

(Used to determine the storage requirement for real element 
parameters as NAELPAR times the number of elements in the 
element group.) 

O 

NAELPRO
P 

Number of real element properties used (determines actual length 
of array AELPROP) 

O 

NDOF1 Number of degrees of freedom per node used by element.  (E.g. 3 
for 2D beam of pipe elements degrees of freedom consist of two 
components of dispacement plus a rotation.) 

O 

NIELPAR Number of integer-values element parameters used for each element. O 
NIELPROP Number of integer element properties used (determines actual 

length of array AELPROP) 
O 

NSD1 Number of coordinates per node used by element. O 

Table AA.2: Description and use of arguments to subroutine GENELKP_nn. 

Variable or 
Array Name 

Description Input/   
Output 

AELPAR Array of real-valued element parameters describing the effect of 
the prior deformation history of the element.  Enters with the 
values of these parameters at the beginning of the loadstep.  
Depending upon the value of IELFLAG these may need to be 
updated. 

I & O 

AELPROP Array of real-valued element properties. I 
ELK Element stiffness matrix, dimensioned as 

ELK(NELDOF,NELDOF). 
O 

ELP Array of element out-of-balance forces.  When multiplied by 
corresponding virtual nodal displacements, one obtains EVW-
IVW, where IVW and EVW are internal and external virtual work 
respectively. 

O 

ELP_L Array containing the derivatives of ELP with respect to the load 
parameter TIME. 

O 

ID Array for which ID(i,j) is the global degree of freedom number for 
nodal degree of freedom i at node j.  Must be used to define the 
LM array, which describes the locations where the element out-of-
balance forces ELP and element stiffness matrix ELK should be 
assembled into the corresponding global arrays for the entire 
system. 

I 

IELPAR Similar to AELPAR, but for integer values element parameters. I & O 
IELPROP Similar to AELPROP, but for interger-valued element properties. I 
IEN Array for which IEN(i) is the global node number corresponding 

to element node number i. 
I 

LM Integer array that describes the location where ELP and ELK is to 
be assembeled into the corresponding global arrays.  Specifically 
LM(i) is the global degree of freedom number corresponding to 
element degree of freedom number I.  This needs to be determined 
using the arrays ID and IEN. 

O 

U Array containing the displacements at the end of the loadstep.  
Specifically U(i,j) is the displacement for nodal degree of freedom I 
at node j, where j is the global node number.  [U(i,IEN(j)) is the 
displacement for nodal degree of freedom I at element node j.] 

I 

UB_L Rate at which the specified portion of the displacementsvi changes 
as a function of of the load parameter TIME.  This is needed for 

I 

                                                 
vi The total nodal displacements can be written as U=UB+Q, where UB represents the known portion of the nodal 

displacements which are specified as a function of time prior to beginning the analysis, and Q represents the 
unknown portion of the displacements.  (As a result, Q is zero for those degrees of freedom where the 
displacements are specified, but on the other hand UB may be non-zero for degrees of freedom where the 
displacement is unknown as well as for those where the displacement is specified.)   UB_L represents the derivative 
of UB with respect to the load parameter TIME.  Furthermore ELP_L represents the derivative of ELP with 
respect to time for a fixed value of the unknown portion of the displacements Q.  This means that for the purpose 
of calculating ELP_L, the derivatives of the nodal displacements with respect to TIME should be obtained from 
the array UB_L. 

Variable or 

Array Name 
Description Input/   

Output 
the calculation of ELP_L.  Stored in the same format as the 
displacements U. 

U0 Array containing the displacements at the beginning of the 
loadstep in the same format as for the displacements U. 

I 

X Array of nodal coordinates, can be used to calculate the geometric 
parameters of the element.  Specifically X(i,j) corresponds to the 
ith nodal coordinate at node j, where j is the global node number.  
Note that the element may but need not make use of all NSD 
nodal coordinates that are present in the analysis. 

I 

IEL Element identification number.  Can be used when printing out 
stresses etc. 

I 

IELFLAG Determines what tasks need to be performed by the 
GENELKP_nn routine.  Specifically IELFLAG this the sum of 
the following quantities: 
2   if  ELK and  ELP_L are to be calculated, 
4   to update the element parameters AELPAR and IELPAR, 
8   to write the stresses and other quantities of interest pertaining 
to the element to unit 6. 
Thus for instance if IELFLAG=0 only ELP, NELDOFI, and LM 
need to be calculated by the GENELKP_nn routine.  If 
IELFLAG=14 then ELK and ELP_L need to be calculated as 
well, the element parameters need to be updated, and output for 
the element needs to be written to unit 6. 

I 

IGRP Element group number.  May be included in the output to unit  6 
when appropriate. 

I 

ISTEP Load step number. I 
NAELPAR Length of array AELPAR. I 
NAELPRO
P 

Length of array AELPROP. I 

NDOF Number of degrees of freedom per node present.  (Not all these 
dof’s need be used by the element.) 

I 

NDOF4BC Used only by special elements, otherwise NDOF4BC=NDOF. I 
NELDOF Maximum number of element degrees of freedom; determines 

declaration of arrays ELK, ELP, etc. 
I 

NELDOFI Number of element degrees of freedom used by element; 
determines how many of the elements of ELK, ELP should be 
assembled into the global arrays. 

O 

NEN Maximum number of nodes used by an element. I 
NIELPAR Length of array IELPAR. I 
NIELPROP Length of array IELPROP I 
NNODES Total number of nodes present in the analysis.  I 
NSD Maximum number of nodal coordinates per node declared for the I 

Variable or 

Array Name 
Description Input/   

Output 
analysis. 

TIME Value of the load parameter at the end of the loadstep. I 
TIME0 Value of the load parameter at the beginning of the loadstep. I 

APPENDIX B. CONCRETE COATING ELEMENT IN FORTRAN 
C 2-node, 2D Concrete Cover Element 
C 
C Deformation is in x-z plane. 
C 
C Nodal Coordinates are: 
C 1) Axial x coordinate, and 
C 2) Transverse z coordinate. 
C 
C Nodal Dof's are: 
C 1) Axial u displacement, 
C 2) Transverse w displacement (in z-direction), and 
C 3) Rotation theta about y-axis (direction by RH rule). 
C 4) Axial slip of concrete over pipe. 
C 
C Real Element Parameters (AELPAR) are: 
C 1:NINTPT) ECMAX = max compressive strain experienced by the concrete 
C NINTPT+1) VSPL(1) = axial plastic slip in tar interface at elmt node 1 
C NINTPT+2) VSPL(2) = axial plastic slip in tar interface at elmt node 1 
C 
C Real Element Properties (AELPROP) are: 
C 1) CDIA = inner diameter of concrete = OD of pipe 
C 2) CTHK = concrete thickness 
C 3) FPC = Concrete compressive strength 
C 4) EPC = concrete strain at max compr stress 
C 5) TAUY = tar interface shear strength 
C 6) SLMOB = tar interface shear mobilisation slip 
C 7) GGP = plastic hardening stiffness for shear-slip relation 
C 8) EPST = thermal axial strain 
C 9) SKAPPAT = thermal curvature 
C 10) BFX = applied force per unit length in x direction 
C 11) BFZ = applied force per unit length in z direction 
C 
C Integer Element Properties (IELPROP) are: 
C 1) NINTPT = number of integration points around half circumference 
C 2) ILFE = load function number for thermal axial strain 
C 3) ILFK = load function number for thermal curvature 
C 4) ILFBF = load function number for body forces 
 
       SUBROUTINE RELPROP_22(AELPROP,IELPROP,     !arrays 
C input/output -->           o       o 
     &   IEXFLAG,IU9STAT,NAELDAT1,NIELDAT1,       !scalars 
C        i       i&o     i        i 
     &   NEN,NAELPAR,NAELPROP,NDOF1,NIELPAR,NIELPROP,NSD1) !scalars 
C        o   o       o        o     o       o        o 
C 
      IMPLICIT REAL*8 (A-H,O-Z) 
      DIMENSION IELPROP(NIELDAT1),AELPROP(NAELDAT1) 
      WRITE(6,'(2X/ 
     &" == Concrete Cover Element (nelmt22) Chosen."/ 
     &"  NINTPT = # int pts around half cicumference")') 
      CALL RDREC1I(NINTPT,'NINTPT',5,IU9STAT) 
      IELPROP(1)=NINTPT 
      NEN=2             !number of element nodes used 
      NAELPAR=NINTPT+2  !number of real element parameters 
      NAELPROP=11       !number of element properties needed 
      NDOF1=4           !number of degrees of freedom per node needed 
      NIELPAR=0         !number of integer element parameters 
      NIELPROP=4        !number of integer element properties needed 
      NSD1=2            !number of coordinates per node used by element 
      CALL CHKINT(NAELPROP,NAELDAT1,'NAELPROP in nelmt22',19) 
      CALL CHKINT(NIELPROP,NIELDAT1,'NIELPROP in nelmt22',19) 
C 
      WRITE(6,'(2X/ 
     &" CDIA = concrete inner diameter = OD of pipe"/ 
     &" CTHK = concrete thickness"/ 
     &" FPC = Concrete compressive strength"/ 
     &" EPC = concrete strain at max compr stress")') 
      CALL RDRECNA(AELPROP(1),'CDIA,CTHK,FPC,EPC',17,IU9STAT,4) 
      WRITE(6,'(2X/ 
     &" TAUY = tar interface shear strength"/ 
     &" SLMOB = tar interface shear mobilisation slip"/ 
     &" GGP = tangent stiffness for interface slip (kin. hard)")') 
      CALL RDRECNA(AELPROP(5),'TAUY,SLMOB,GGP',14,IU9STAT,3) 



EP 2006-5161 - 20 - Unrestricted 
 

C 
      WRITE(6,'(2X/ 
     &" Initial axial strain and curvature are given by"/ 
     &"    EPST*ALOADF(TIME,ILFE,0), and"/ 
     &"    SKAPPAT*ALOADF(TIME,ILFK,0), respectively.")') 
      CALL RDRECNA(AELPROP(8),'EPST,SKAPPAT',12,IU9STAT,2) 
      CALL RDRECNI(IELPROP(2),'ILFE,ILFK',9,IU9STAT,2) 
C 
      WRITE(6,'(2X/ 
     &" Input body forces (per unit length) acting on element,"/ 
     &"   BFX = force per unit length in x-direction"/ 
     &"   BFZ = force per unit length in z-direction"/ 
     &"   ILFBF = time function number associated with body force, or"/ 
     &"           zero for constant body force acting at all times.")') 
      CALL RDRECNA1I(AELPROP(10),IELPROP(4),'BFX,BFZ,ILFBF', 
     & 13,IU9STAT,2) 
      RETURN 
      END 
C End of subroutine RELPROP_22 
C 
C================================================================================
==== 
C 
      SUBROUTINE GENELKP_22(AELPAR,AELPROP, 
     &   ELK,ELP,ELP_L, 
     &   ID,IELPAR,IELPROP, 
     &   IEN,LM, 
     &   U,UB_L,U0,X,  
     &   IEL,IELFLAG,IGRP,ISTEP,NAELPAR,NAELPROP,NDOF,NDOF4BC, 
     &   NELDOF,NELDOFI, 
     &   NEN,NIELPAR,NIELPROP,NNODES,NSD,TIME,TIME0)  
C 
      IMPLICIT REAL*8 (A-H,O-Z) 
C 
      DIMENSION AELPAR(NAELPAR),AELPROP(NAELPROP), 
     & ELK(NELDOF,NELDOF),ELP(NELDOF),ELP_L(NELDOF), 
     & ID(NDOF,NNODES),IELPAR(NIELPAR),IELPROP(NIELPROP), 
     & IEN(NEN),LM(NELDOF), 
     & U(NDOF,NNODES),UB_L(NDOF,NNODES),U0(NDOF,NNODES),X(NSD,NNODES) 
C 
C Local Declarations: 
      LOGICAL LF_INDIC 
      DIMENSION ELKB(6,6),IDB(8) 
      DATA IDB/1,2,3,1,4,5,6,4/ 
      PARAMETER(PI=3.14159265359D0,DEG=0.01745329251994D0) 
C 
      DIMENSION UEL(8),UELB_L(8),D(3,3),DB(3,6),GGT(2),TAU(2) 
C 
C Initialise element parameters, if appropriate: 
      IF(LF_INDIC(IELFLAG,6)) THEN 
        IF(NAELPAR.GT.0) CALL CLEAR(AELPAR,NAELPAR) 
        IF(NIELPAR.GT.0) CALL ICLEAR(IELPAR,NIELPAR) 
        IF(IELFLAG.EQ.2**6) RETURN 
      ENDIF 
C 
C Form LM,  UELB_L, and NELDOFI: 
      J=0 
      DO I=1,2 
        INODE=IEN(I) 
        DO II=1,4 
          J=J+1 
          LM(J)=ID(II,INODE) 
          UEL(J)=U(II,INODE) 
          UELB_L(J)=UB_L(II,INODE) 
        ENDDO 
      ENDDO                 
      NELDOFI=J 
C 
C Geometry & Displacement derivatives 
      XX=0.5D0*(X(1,IEN(2))+X(1,IEN(1))) 
      CL=X(1,IEN(2))-X(1,IEN(1)) 
      CZP=(  X(2,IEN(2))-X(2,IEN(1))  )/CL 
      UP=(UEL(5)+UEL(8)-UEL(1)-UEL(4))/CL 
      WP=(UEL(6)-UEL(2))/CL 
      TP=(UEL(7)-UEL(3))/CL 
      SZP=CZP+WP 

C 
C Recover Element Properties: 
      CDIA=AELPROP(1) 
      CTHK=AELPROP(2) 
      FPC=AELPROP(3) 
      EPC=AELPROP(4) 
      TAUY=AELPROP(5) 
      SLMOB=AELPROP(6) 
      GGP=AELPROP(7) 
      EPST=AELPROP(8) 
      SKAPPAT=AELPROP(9) 
      BFX=AELPROP(10) 
      BFZ=AELPROP(11) 
      NINTPT=IELPROP(1) 
      ILFE=IELPROP(2) 
      ILFK=IELPROP(3) 
      ILFBF=IELPROP(4) 
C 
C Section Strains & time deriv for const nodal displ. 
      EPS=UP+(CZP+0.5D0*WP)*WP 
      SKAPPA=TP 
      EPS=EPS-EPST*ALOADF(TIME,ILFE,0) 
C     VS=0.5D0*(UEL(4)+UEL(8)) !relative axial slip 
      EPS_L=-EPST*ALOADF(TIME,ILFE,1) 
      SKAPPA=SKAPPA-SKAPPAT*ALOADF(TIME,ILFK,0) 
      SKAPPA_L=-SKAPPAT*ALOADF(TIME,ILFK,1) 
C 
C Stress Resultants, CN, CM & derivatives D(i,j) wrt EPS & SKAPPA: 
      CN=0.D0 
      CM=0.D0 
      CALL CLEAR(D,9) 
      DTHETA=NINTPT-1 
      DTHETA=PI/DTHETA 
      RR=0.5D0*(CDIA+CTHK) 
      TRXSA1=RR*DTHETA*CTHK 
      TRXSA2=TRXSA1+TRXSA1 
      EEC=2.D0*FPC/EPC !elastic modulus of concrete 
c     EPC2=EPC+EPC 
      DO I=1,NINTPT 
        THETA=I-1   
        THETA=THETA*DTHETA  !theta = 0 at 12 o'clock 
        ZZ=RR*COS(THETA) 
        EPSX=EPS+ZZ*SKAPPA 
        ECMAX0=AELPAR(I) 
        ECMAX1=MAX(ECMAX0,-EPSX) 
C   Calc conc stress (SC), tangent modulus (ETC): 
        IF(EPSX.GT.0.D0) THEN !cracks open 
          SC=0.D0 
          ETC=0.D0 
        ELSE IF(ECMAX1.GT.ECMAX0) THEN 
          CALL SECONC1(FCMAX1,ETC,ECMAX1,FPC,EPC,EEC) 
          SC=-FCMAX1 
          IF(LF_INDIC(IELFLAG,2)) AELPAR(I)=ECMAX1 
        ELSE 
          CALL SECONC1(FCMAX0,ETC,ECMAX0,FPC,EPC,EEC) 
          SCEL=-FCMAX0+EEC*(EPSX+ECMAX0) 
          IF(SCEL.GT.0.D0) THEN 
            SC=0.D0 
            ETC=0.D0 
          ELSE 
            SC=SCEL 
            ETC=EEC 
          ENDIF 
        ENDIF 
C 
        IF(LF_INDIC(IELFLAG,3)) THEN 
          WRITE(6,'(" X,THETA,EPSX,SC,ECMAX",5G13.5)') 
     &           XX,THETA/DEG,EPSX,SC,ECMAX1 !theta in deg 
        ENDIF 
        IF(I.EQ.1.OR.I.EQ.NINTPT) THEN 
          SC=SC*TRXSA1 
          ETC=ETC*TRXSA1 
        ELSE 
          SC=SC*TRXSA2 
          ETC=ETC*TRXSA2 
        ENDIF 

        CN=CN+SC 
        CM=CM+SC*ZZ 
        D(1,1)=D(1,1)+ETC 
        TST1=ETC*ZZ 
        D(1,2)=D(1,2)+TST1 
        D(2,2)=D(2,2)+TST1*ZZ 
      ENDDO 
      D(2,1)=D(1,2) 
      CN_L=D(1,1)*EPS_L+D(1,2)*SKAPPA_L 
      CM_L=D(2,1)*EPS_L+D(2,2)*SKAPPA_L 
C 
C Calc shear stress in tar (TAU), tangent shear stiffness (GGT); update VSPL: 
C   The basis for this is a kinematic hardening model in which 
C   TAU=GGE (VS-VSPL) 
C   TAUyield=GGP*VSPL+/-TAUY 
C 
      GGE=TAUY/SLMOB            !elastic stiffness 
      DO I=1,2 
        VSPL=AELPAR(NINTPT+I)     !plastic slip 
        VS=UEL(4*I) 
        TAUE=GGE*(VS-VSPL)        !elastic predictor stress 
        TAUSH=GGP*VSPL 
        TAUYP=TAUSH+TAUY          !yield stress for the positive direction 
        TAUYN=TAUSH-TAUY          !yield stress for the negative direction 
        IF(TAUE.GT.TAUYP) THEN    !yielding in the positive direction 
          GGT(I)=GGP*GGE/(GGE+GGP)     !tangent stiffness 
          TAU(I)=TAUYP+GGT(I)*(TAUE-TAUYP)/GGE    !current stress 
          VSPL=VS-TAU(I)/GGE                   !current plastic slip 
        ELSE IF(TAUE.LT.TAUYN) THEN  !yielding in the negative direction 
          GGT(I)=GGP*GGE/(GGE+GGP)     !tangent stiffness 
          TAU(I)=TAUYN+GGT(I)*(TAUE-TAUYN)/GGE    !current stress 
          VSPL=VS-TAU(I)/GGE                   !current plastic slip 
        ELSE                       !elastic behaviour 
          TAU(I)=TAUE 
          GGT(I)=GGE 
        ENDIF 
        IF(LF_INDIC(IELFLAG,2)) AELPAR(NINTPT+I)=VSPL 
      ENDDO 
C         
      TRAREA=0.5D0*PI*CDIA*CL 
C                              
C Compute Element out-of-balance load vector: 
      CQ1=SZP*CN 
      TST=0.5D0*CL*ALOADF(TIME,ILFBF,0) 
      BFX1=TST*BFX 
      BFZ1=TST*BFZ 
      ELP(1)=CN+BFX1 
      ELP(2)=CQ1+BFZ1 
      ELP(3)=CM 
      ELP(4)=ELP(1)-TAU(1)*TRAREA 
      ELP(5)=-CN+BFX1 
      ELP(6)=-CQ1+BFZ1 
      ELP(7)=-CM 
      ELP(8)=ELP(5)-TAU(2)*TRAREA 
C 
      IF(LF_INDIC(IELFLAG,1)) THEN  !compute ELK and ELP_L  
C 
C   Form DB(I,J) =  SUM(  D(I,K) B(K,J)  ,K=1,3) 
        DO I=1,3 
          TST1=D(I,1)           
          TST2=SZP*D(I,1)+D(I,3)    
          TST3=D(I,2) 
          TST1=TST1/CL 
          TST2=TST2/CL 
          TST3=TST3/CL 
          TST4=D(I,3)/2.D0 
          DB(I,1)=-TST1 
          DB(I,2)=-TST2 
          DB(I,3)=-TST3+TST4 
          DB(I,4)=+TST1 
          DB(I,5)=+TST2 
          DB(I,6)=+TST3+TST4 
        ENDDO  
C 
C Form ELKB(I,J) = SUM(  B(K,I)*DB(K,J)   ,K=1,3) 
        DO J=1,6 

          TST1=DB(1,J)           
          TST2=SZP*DB(1,J)+DB(3,J)    
          TST3=DB(2,J)          
          TST4=DB(3,J)*CL/2.D0  
          ELKB(1,J)=-TST1 
          ELKB(2,J)=-TST2 
          ELKB(3,J)=-TST3+TST4 
          ELKB(4,J)=+TST1 
          ELKB(5,J)=+TST2 
          ELKB(6,J)=+TST3+TST4 
        ENDDO 
C Add geometric stiffness matrix 
        TST1=CN/CL 
        ELKB(2,2)=ELKB(2,2)+TST1 
        ELKB(2,5)=ELKB(2,5)-TST1 
        ELKB(5,2)=ELKB(5,2)-TST1 
        ELKB(5,5)=ELKB(5,5)+TST1 
C Transform to new dof's & add contribution from slip stiffness: 
        DO I=1,8 
          IB=IDB(I) 
          DO J=1,8 
            JB=IDB(J) 
            ELK(I,J)=ELKB(IB,JB) 
          ENDDO 
        ENDDO 
        ELK(4,4)=ELK(4,4)+GGT(1)*TRAREA 
        ELK(8,8)=ELK(8,8)+GGT(2)*TRAREA 
C Compute ELP_L: 
        CQ1=SZP*CN_L 
        TST=0.5D0*CL*ALOADF(TIME,ILFBF,1) 
        BFX1=BFX*TST 
        BFZ1=BFZ*TST 
        ELP_L(1)=CN_L+BFX1 
        ELP_L(2)=CQ1+BFZ1 
        ELP_L(3)=CM_L 
        ELP_L(4)=ELP_L(1) 
        ELP_L(5)=-CN_L+BFX1 
        ELP_L(6)=-CQ1+BFZ1 
        ELP_L(7)=-CM_L 
        ELP_L(8)=ELP_L(5) 
        DO I=1,NELDOFI 
          TST1=ELP_L(I) 
          DO J=1,NELDOFI 
            TST1=TST1-ELK(I,J)*UELB_L(J) 
          ENDDO 
          ELP_L(I)=TST1 
        ENDDO 
      ENDIF 
      IF(LF_INDIC(IELFLAG,3)) THEN !print stresses in element 
        WRITE(6,1100) IEL,XX,CN,CM 
        WRITE(6,1200) X(1,IEN(1)),(-ELP(I),I=1,4) 
        WRITE(6,1300) X(1,IEN(2)),(-ELP(I),I=5,8) 
      ENDIF 
      RETURN 
1100  FORMAT(' IEL=',I7,'  X=',G12.4,'  N=',G12.4,'  M=',G12.4) 
1200  FORMAT(' X=',G12.4,'  S1:S4=',4G12.4) 
1300  FORMAT(' X=',G12.4,'  S5:S8=',4G12.4) 
      END 
C 
      SUBROUTINE SECONC1(FC,ETC,EC,FPC,EPC,EEC) 
C   input/output         o  o   i  i   i   i 
C Defines backbone stress-strain relation for concrete in compression. 
C Parabolic stress-strain relation up to ultimate; const stress thereafter. 
C FC = concrete stress (positive for compression) 
C ETC = tangent modulus 
C EC = strain in concrete (positive for compression) 
C FPC = concrete compressive strength 
C EPC = concrete strain at ultimate compression (positive) 
C EEC = 2 FPC / EPC = concrete elastic modulus 
      IMPLICIT REAL*8 (A-H,O-Z) 
      XI=(EPC-EC)/EPC 
      IF(XI.GT.0.D0) THEN 
        FC=FPC*(1.D0-XI*XI) 
        ETC=EEC*XI 
      ELSE 
        FC=FPC 

        ETC=0.D0 
      ENDIF 
      RETURN 
      END 
C 
C Log of Developments 
C Derived from nelmt18 (mod defl elastic beam element) on 2 Oct 2001. 
C nelmt18 appears to have been derived from nelmt04 (mod defl pipe element) 
C 12 Feb 2006: adding a small stiffness against slip, because of convergence 
C   difficulties encountered without this when none of the slip dof's within 
C   a pipe joint are restrained. Input files require adding GGT after SLMOB 
C   (on same line) as part of the element property input. 
C 12 Feb 2006: changing number of integration points for slip work from 
C   1 in the middle of the element to 2, with one at each node. 
C   Otherwise zero energy deformation modes form when all the concrete 
C   is cracked in tension.  This rather than the lack of stiffness 
C   against slip seems to have been the reason for convergence difficulties. 

